

Calculating Physical and Mental Health Summary Scores for PROMIS-29 v2.0 and v2.1

Karen L. Spritzer and Ron D. Hays

The PROMIS-29 Profile measure assesses pain intensity using a single 0–10 numeric pain rating item and seven health domains using four items each: physical functioning, fatigue, pain interference, depression, anxiety, ability to participate in social roles and activities, and sleep disturbance. This memo describes scoring of the PROMIS-29 into two summary scores (physical and mental health) using SAS software. This scoring is valid for versions 2.0 and 2.1. For supporting psychometric analyses, please see Hays, Spritzer, Schalet, and Cella (2018). The PDF version of the PROMIS 29 v2.1 is available on http://www.healthmeasures.net/index.php?option=com_instruments&view=measure&id=849&Itemid=992.

To score the PROMIS-29 Physical Health and Mental Health Summary Scales, users need to ensure that each domain has an associated T-score (except for the numerical pain rating scale). In the example we provide below, the PROMIS-29 Profile domain-level T-scores are obtained using the HealthMeasures Scoring Service, powered by Assessment CenterSM (https://www.assessmentcenter.net/ac_scoring-service). Other options for obtaining domain-level T-scores are describe on <http://www.healthmeasures.net/score-and-interpret/calculate-scores>. This includes administration the PROMIS-29 in electronic platforms such as REDCap that produce domain-level T-scores.

In our example, we used the HealthMeasures Scoring Service and SAS software, Version 9.4 (SAS Institute, Inc., Cary, NC, USA) to create an extract of our data (renaming/recoding items when necessary to conform to PROMIS item structure. As seen in Appendix A, “export_AC.sas” creates a comma delimited file “prom29.csv”. Next, we uploaded prom29.csv to the HealthMeasures Scoring Service naming the instrument “prom29_1” and using the default calibration sample (PROMIS Wave 1). “prom29_1.log” (log of information regarding how the data was read and analyzed) and “prom29_1.csv” (output file containing scored results) files were returned to us by email.

As shown in Appendix B, the final step shows score29.sas reading prom29_1.csv and restructuring it

to create one record for each PIN/Assessment with 7 domain scores. It then merges in the 0-10 raw pain intensity item and creates a z-score using the PROMIS Wave 1 general population mean (2.31) and SD (2.34) to create PAIN1_Z. Next, it creates two new composites: PAINAV_Z is the average of this new pain intensity z-score (PAIN1_Z) and the pain interference z-score (PAIN4A_Z), and creates EMOT_Z as the average of the anxiety (ANX4A_Z) and depression (DEP4A_Z) z-scores.

The PROMIS-29 physical health and mental health Z-scores are then calculated as the sum of the product of the factor scoring coefficients (reference below) with the Z-scores.

/ multiply z-scores with corresponding factor scoring coefficients and sum */*

```
ph29_z=(phy4a_z * 0.872) +  
  (painav_z * -0.094) +  
  (social4a_z * 0.113)  
  + (fatig4a_z * -  
  0.009) + (slpd4a_z *  
  0.002) + (emot_z *  
  0.003);
```

```
mh29_z=(phy4a_z * -0.015) +  
  (painav_z * -0.154) +  
  (social4a_z * 0.252)  
  + (fatig4a_z * -  
  0.351) + (slpd4a_z *  
  -0.139) + (emot_z *  
  -0.257);
```

References

Cella, D., Gershon, R., Bass, M., & Rothrock, N. (2017). *Assessment Center Scoring ServiceSM User Manual* v1.4. Accessed August 2018 at

https://www.assessmentcenter.net/ac_scoring/service/templates/UserManual.pdf.

Cella, D., Riley, W., Stone, A., Rothrock, N., Reeve, B., Young, S., Amtmann, D., Bode, R., Buysse, D., Choi, S., Cook, K., DeVellis, R., DeWalt, D., Fries, J. F., Gershon, R., Hahn, E. A., Pilkonis, P., Revicki, D., Rose, M., Weinfurt, K., Lai, J., & Hays, R. D. (2010). Initial item banks and first wave testing of the Patient- Reported Outcomes Measurement Information System (PROMIS) network: 2005-2008. *Journal of Clinical Epidemiology*, 63 (11), 1179-1194.

Hays, R. D., Spritzer, K. L., Schalet, B. D., & Cella, D. (2018). PROMIS®-29 v2. 0 profile physical and mental health summary scores. *Quality of Life Research*, 27, 1885-1891.

Appendix A: HealthMeasures Scoring Service Phase

export_AC.sas: SAS code to create extract for the HealthMeasures Scoring Service (creates prom29.csv).

```
options ls=132 ps=54 nocenter;

libname library "C:\projects\PROMIS\AC\scoring_example";

/*****/
TITLE "PROMIS 2010 Centering"; run;
DATA prom29;
  set library.ReCenter_14_08_2013V1Comp;
/*****/

Assmnt=1;

* rename items for use by Assessment Center/ reverse score items when necessary using PROMIS29_v2.pdf Profile as
a guide;

rename
rid          = pin

QPFA11      = PFA11
QPFA21      = PFA21
QPFA23      = PFA23
QPFA53      = PFA53

QEDANX01    = EDANX01
QEDANX40    = EDANX40
QEDANX41    = EDANX41
QEDANX53    = EDANX53

QEDDEP04    = EDDEP04
QEDDEP06    = EDDEP06
QEDDEP29    = EDDEP29
QEDDEP41    = EDDEP41

QHI7        = HI7
QAN3        = AN3
QFATEXP41   = FATEXP41
QFATEXP40   = FATEXP40

QSleep20    = Sleep20
QSleep44    = Sleep44

QSRPPER11CaPS = SRPPER11_CaPS
QSRPPER18CaPS = SRPPER18_CaPS
QSRPPER23CaPS = SRPPER23_CaPS
QSRPPER46CaPS = SRPPER46_CaPS

QPAININ9    = PAININ9
QPAININ22   = PAININ22
QPAININ31   = PAININ31
QPAININ34   = PAININ34;

* item reversal;
sleep109    = 6 - Qsleep109;
sleep116    = 6 - Qsleep116;

drop Qsleep109 Qsleep116;
run;

/** check scoring **/

proc corr alpha nomiss data=prom29; var pfa11 pfa21 pfa23 pfa53; run;
proc corr alpha nomiss data=prom29; var edanx01 edanx40 edanx41 edanx53; run;
proc corr alpha nomiss data=prom29; var eddep04 eddep06 eddep29 eddep41; run;
proc corr alpha nomiss data=prom29; var hi7 an3 fatexp41 fatexp40; run;
proc corr alpha nomiss data=prom29; var sleep109 sleep116 sleep20 sleep44; run;
proc corr alpha nomiss data=prom29; var srpper11_caps srpper18_caps srpper23_caps srpper46_caps; run;
proc corr alpha nomiss data=prom29; var painin9 painin22 painin31 painin34; run;
```

Appendix A continued: HealthMeasures Scoring Service Phase

```

/* convert to character to change SAS missing . to blanks for use in AC (AC requires missing to be blank or
value of 'SKIP' - refer to AC manual:
https://www.assessmentcenter.net/ac_scoringsservice/templates/UserManual.pdf) */

data prom29; length i1-i28 $1.; set prom29;
array invar pfall1 pfa21 pfa23 pfa53
  edanx01 edanx40 edanx41 edanx53
  eddep04 eddep06 eddep29 eddep41
  hi7 an3 fatexp41 fatexp40
  sleep109 sleep116 sleep20 sleep44
  srpper11_caps srpper18_caps srpper23_caps srpper46_caps
  painin9 painin22 painin31 painin34;
array outvar $ i1-i28;

do over invar;
if invar > . then outvar=put(invar,$1.);
else outvar=" ";
end;

run;

TITLE "Check a few recodes" ; run;
proc freq data=prom29; tables pfall1*i1 srpper11_caps*i21 srpper18_caps*i22 srpper23_caps*i23
  srpper46_caps*i24 /list missing; run;

/* output to csv */
data _null_; file 'prom29.csv' dlm=','; set prom29;
if _N_=1 then put @1
"PIN,Assmnt,pfall1,pfa21,pfa23,pfa53,edanx01,edanx40,edanx41,edanx53,eddep04,eddep06,eddep29,eddep41,hi7,an3,fate
xp41,fatexp40,
sleep109,sleep116,sleep20,sleep44,srpper11_caps,srpper18_caps,srpper23_caps,srpper46_caps,painin9,paini
n31,painin34";

put @1 PIN Assmnt i1-i28;
run;

```

prom29.csv: (first case as input into AC)

```

PIN,Assmnt,pfall1,pfa21,pfa23,pfa53,edanx01,edanx40,edanx41,edanx53,eddep04,eddep06,eddep29,eddep41,hi7,an3,fate
xp41,fatexp40,sleep109,sleep116,sleep20,sleep44,srpper11_caps,srpper18_caps,srpper23_caps,srpper46_caps,painin9,p
ainin22,painin31,painin34
24,1,5,5,5,2,2,2,2,1,1,1,1,1,1,1,1,2,2,2,1,5,5,5,1,1,1,1

```

prom29_1_log.csv: log of AC run (not shown)

prom29_1.csv: (first case) scores produced from AC

Report Generated: 1/23/2018 5:43:37 PM								
Original Instrument: PROMIS-29 Profile v2.0								
Calibration Sample: Promis Wave 1								
PIN	Assmnt	Inst	RawScore	Theta	TScore	SE	ScrdCnt	ItmCnt
24	1	prom29_1 - Pain Interference	4	-0.84	41.6	6.1	4	28
24	1	prom29_1 - Depression/Sadness	4	-0.9	41	6.2	4	28
24	1	prom29_1 - Physical Function	20	0.69	56.9	6.6	4	28
24	1	prom29_1 - Ability to Participate in Social Roles	20	1.42	64.2	5.1	4	28
24	1	prom29_1 - Fatigue	4	-1.63	33.7	4.9	4	28
24	1	prom29_1 - Anxiety/Fear	8	0.6	56	2.4	4	28
24	1	prom29_1 - Sleep Disturbance	7	-0.58	44.2	3.3	4	28

Appendix B: SAS Scoring Phase

score29.lst:

Assmnt	N Obs	Variable	Label	N	Mean	Std Dev	Minimum	Maximum
1	1000	ph29_t	PROMIS29 v2.0 PH summary (T)	999	49.40	8.94	22.19	58.63
		mh29_t	PROMIS29 v2.0 MH summary (T)	999	49.71	8.86	20.33	65.51
		phy4a_t	Physical Function (T)	1000	49.18	8.61	22.90	56.90
		painav_t	pain composite (T)	1000	52.20	9.48	40.86	79.23
		social4a_t	Ability to Part in Soc Roles/Activ (T)	1000	52.58	9.21	27.50	64.20
		fatig4a_t	Fatigue (T)	999	49.91	9.62	33.70	75.80
		slpd4a_t	Sleep Disturbance (T)	1000	50.98	8.55	32.00	73.30
		emot_t	anxiety/dep composite (T)	1000	52.00	9.13	40.65	80.35
		anx4a_t	Anxiety/Fear (T)	1000	52.91	9.54	40.30	81.40
		dep4a_t	Depression/Sadness (T)	1000	51.10	9.66	41.00	79.30
		pain4a_t	Pain Interference (T)	999	51.23	9.25	41.60	75.60
		pain1_t	Pain intensity (T)	1000	53.18	11.11	40.13	82.86

score29.sas:

```

options ls=132 ps=54 nocenter;
*-----;
libname AC "C:\Projects\PROMIS\AC\scoring_example\";
/*****/
* AC\prom29_1.csv: first 1000 records scored in AC from prom29.csv (uniform # of cases);
* AC\testing\prom29_1.csv: modified prom29.csv first 1000 records illustrate variable number of rows/pin;
* AC\testing\prom29_2.csv: modified prom29.csv first 1000 records illustrate 2 assessments;
/*****/
PROC IMPORT OUT= WORK.AC
DATAFILE= "AC\testing\prom29_1.csv" DBMS=CSV REPLACE;
    GETNAMES=NO;
    DATAROW=6;
RUN;
/*****/
data ac; set ac;
rename
var1=pin
var2=assmnt
var3=inst
var6=Tscore;

*rename
var1=pin
var2=assmnt
var3=inst
var4=RawScore
var5=Theta
var6=Tscore
var7=SE
var8=ScrdCnt
var9=ItmCnt;
keep var1 var2 var3 var6;
run;
/*
TITLE "Initial read from AC CSV file"; run;
proc freq; tables inst; run;
*/
/*****/

```

Appendix B continued: SAS Scoring Phase

```

/*****/
* restructure data: one record per PIN/Assmnt;
proc sort data=ac; by pin assmnt; run;
data ac(keep=pin assmnt pain4a_t dep4a_t phy4a_t social4a_t fatig4a_t anx4a_t slpd4a_t);
    set ac; by pin assmnt;
if first.assmnt then do;
pain4a_t =.;
dep4a_t =.;
phy4a_t =.;
social4a_t=.;
fatig4a_t =.;
anx4a_t =.;
slpd4a_t =.;
end;

retain pin assmnt pain4a_t dep4a_t phy4a_t social4a_t fatig4a_t anx4a_t slpd4a_t;

if index(inst,"Pain Interference") > 0 then pain4a_t =Tscore;
if index(inst,"Depression/Sadness") > 0 then dep4a_t =Tscore;
if index(inst,"Physical Function") > 0 then phy4a_t =Tscore;
if index(inst,"Ability to Participate in Social Roles/Activities") > 0 then social4a_t=Tscore;
if index(inst,"Fatigue") > 0 then fatig4a_t =Tscore;
if index(inst,"Anxiety/Fear") > 0 then anx4a_t =Tscore;
if index(inst,"Sleep Disturbance") > 0 then slpd4a_t =Tscore;

if last.assmnt then output;
run;
/*****/

/*****/
* pick up single-item 0-10 pain intensity item from original dataset;
libname library "C:\projects\PROMIS\AC\scoring_example";
DATA prom;
    set library.ReCenter_14_08_2013V1Comp(keep=rid qglobal07);
Assmnt=1;
rename
    rid = pin
    qglobal07=pain_average;
run;
proc sort data=prom; by pin assmnt; run;
/*****/

/*****/
data all;
    merge prom(keep=pin assmnt pain_average)
          ac(keep=pin assmnt pain4a_t dep4a_t phy4a_t social4a_t fatig4a_t anx4a_t slpd4a_t);
by pin assmnt;

* create Z-scores from T;
pain4a_z = (pain4a_t - 50)/10;
dep4a_z = (dep4a_t - 50)/10;
phy4a_z = (phy4a_t - 50)/10;
social4a_z = (social4a_t - 50)/10;
fatig4a_z = (fatig4a_t - 50)/10;
anx4a_z = (anx4a_t - 50)/10;
slpd4a_z = (slpd4a_t - 50)/10;

* create Z-score for pain_average using PROMIS Wave 1 general population mean=2.31, sd=2.34;
pain1_z = (pain_average - 2.31)/2.34;

*compute composite for pain interference and pain intensity using Z-scores;
painav_z = mean(of pain4a_z pain1_z);

* compute composite for emotion using anxiety and depression Z-scores;
emot_z = mean(of anx4a_z dep4a_z);

```

Appendix B continued: SAS Scoring Phase

```
* create T-score versions;
pain1_t = 50 + (pain1_z * 10);
painav_t = 50 + (painav_z *
10); emot_t = 50 + (emot_z *
10);

label
painav_z ="pain composite (Z)"
emot_z ="anxiety/dep composite
(Z)" painav_t ="pain composite (T)"
emot_t ="anxiety/dep composite (T)"

social4a_z="Ability to Part in Soc Roles/Activ
(Z)" anx4a_z ="Anxiety/Fear (Z)"
dep4a_z ="Depression/Sadness
(Z)" fatig4a_z ="Fatigue (Z)"
pain4a_z ="Pain Interference (Z)"
phy4a_z ="Physical Function (Z)"
slpd4a_z ="Sleep Disturbance (Z)"
pain1_z ="Pain intensity (Z)"

social4a_t="Ability to Part in Soc Roles/Activ
(T)" anx4a_t ="Anxiety/Fear (T)"
dep4a_t ="Depression/Sadness
(T)" fatig4a_t ="Fatigue (T)"
pain4a_t ="Pain Interference (T)"
phy4a_t ="Physical Function (T)"
slpd4a_t ="Sleep Disturbance (T)"
pain1_t ="Pain intensity (T)"
;

/* factor scoring coefficients from "PROMIS-29 V2.0 Physical and
Mental Health Summary Scores" - Hays/Spritzer/Schalet/Cella

ph_phyfun= 0.872 ; mh_phyfun= -0.015 ;
ph_painav= -0.094 ; mh_painav= -0.154 ;
ph_social= 0.113 ; mh_social= 0.252 ;
ph_fatig = -0.009 ; mh_fatig = -0.351 ;
ph_sleep = 0.002 ; mh_sleep = -0.139 ;
ph_emot = 0.003 ; mh_emot = -0.257 ;
*/

/* multiply z-scores with corresponding factor scoring coefficient */
ph29_z=(phy4a_z * 0.872) +
(painav_z * -0.094) +
(social4a_z * 0.113) +
(fatig4a_z * -0.009) +
(slpd4a_z * 0.002) +
(emot_z * 0.003);

mh29_z=(phy4a_z * -0.015) +
(painav_z * -0.154) +
(social4a_z * 0.252) +
(fatig4a_z * -0.351) +
(slpd4a_z * -0.139) +
(emot_z * -0.257);

* create T-score versions: ;
ph29_t= 50 + (ph29_z * 10);
mh29_t= 50 + (mh29_z * 10);
label
ph29_z="PROMIS29 v2.0 PH summary (Z)"
mh29_z="PROMIS29 v2.0 MH summary (Z)"
ph29_t="PROMIS29 v2.0 PH summary (T)"
mh29_t="PROMIS29 v2.0 MH summary
(T)"; run;

TITLE "Score29 means"; run;
proc means data=all maxdec=2;
var ph29_t mh29_t phy4a_t painav_t social4a_t fatig4a_t slpd4a_t emot_t
anx4a_t dep4a_t pain4a_t pain1_t ;
class assmnt;
```


run